

# Module 6

## Differentiëren, integreren, sommatie, limieten

---

|                     |                                                                                             |
|---------------------|---------------------------------------------------------------------------------------------|
| <b>Onderwerp</b>    | Differentiëren, integreren en berekenen van limieten van functies; sommatie                 |
| <b>Voorkennis</b>   | VWO-stof over differentiëren, integreren en limieten                                        |
| <b>Expressies</b>   | diff, Diff, \$, D, int, Int, value, Parts, Combine, Change, :-, limit, Limit, sum, Sum, add |
| <b>Bibliotheken</b> | IntegrationTools                                                                            |
| <b>Zie ook</b>      | Module 1, 3, 7.                                                                             |

---

## 6.1 Inleiding

De Maple-commando's voor het berekenen van afgeleiden, integralen, sommen en limieten (respectievelijk `diff`, `int`, `sum` en `limit`) hebben een groot aantal eigenschappen gemeenschappelijk. We behandelen ze daarom hier achter elkaar.

Veronderstel dat  $f$  een functie is, bijvoorbeeld  $f(x) = x^2$ . Deze functie kan in Maple worden ingevoerd

- als *expressie* waarin de variabele  $x$  voorkomt: `fx := x^2`
- als *functie* (of procedure): `f := t -> t^2`

(als  $x \in \mathbb{N}$  spreken we liever van een *rij* in plaats van een functie). In deze module hanteren we de volgende

**Afspraak:** We veronderstellen dat `f` een procedure is en `fx` een gewone expressie waarin de variabele `x` voorkomt.

## 6.2 De afgeleide

Als  $f$  een functie van één variabele is, dan is  $f$  *differentieerbaar* in het punt  $x$  als

$$\lim_{h \rightarrow 0} \frac{f(x+h) - f(x)}{h}$$

bestaat. De waarde van deze limiet wordt aangeduid met  $f'(x)$  of  $\frac{d}{dx} f(x)$  (eventueel  $\frac{df}{dx}(x)$ ) en heet: de *afgeleide* van  $f$  in het punt  $x$ . De functie  $f'$  heet de *afgeleide functie* van  $f$ .

Als de afgeleide functie differentieerbaar is, dan bestaat dus de afgeleide van de functie  $f'$  en heet de tweede afgeleide van  $f$ . Deze wordt genoteerd als  $f''(x)$  of  $\frac{d^2}{dx^2} f(x)$ .

Voor het berekenen van afgeleiden kan gebruik gemaakt worden van de volgende rekenregels:

$$\begin{aligned} \text{Productregel:} \quad & \frac{d}{dx} (f(x)g(x)) = f(x)g'(x) + f'(x)g(x) \\ \text{Kettingregel:} \quad & \frac{d}{dx} f(g(x)) = f'(g(x))g'(x). \end{aligned}$$

Voor functies van twee of meer variabelen definieert men de *partiële afgeleiden* als

$$\begin{aligned} \frac{\partial}{\partial x} f(x, y) &= \lim_{h \rightarrow 0} \frac{f(x+h, y) - f(x, y)}{h}, \\ \frac{\partial}{\partial y} f(x, y) &= \lim_{h \rightarrow 0} \frac{f(x, y+h) - f(x, y)}{h}. \end{aligned}$$

**Impliciet differentiëren.** Door een *vergelijking* van de vorm  $f(x, y) = c$ , met  $c$  een constante wordt (bijvoorbeeld)  $y$  als differentieerbare functie van  $x$  vastgelegd.<sup>12</sup> Als we  $y(x)$  schrijven voor deze functie, dan geldt dat

$$F(x) = f(x, y(x)) = c \quad \text{voor alle } x.$$

Dus  $F$  is een constante functie, zodat  $F'(x) = 0$  voor alle  $x$ . Met behulp van de kettingregel kan zo een de afgeleide  $y'(x)$  worden uitgedrukt in de variabelen  $x$  en  $y$ , zonder  $y(x)$  expliciet te berekenen.

## 6.3 Differentiëren

**diff**

**Expressies.** Met het commando `diff(fx, x)` kan men de afgeleide van een expressie berekenen. Het resultaat is weer een expressie.

**\$**

Ook hogere afgeleiden kunnen worden bepaald. Bijvoorbeeld  $f''(x)$  met `diff(fx, x, x)`, en  $f'''(x)$  via `diff(fx, x, x, x)`, of, korter, via `diff(fx, x$3)`. De expressie `x$3` is een afkorting voor `x, x, x`. Zie ook Module 8. Met de `$`-constructie kan men Maple zelfs de  $n^{\text{de}}$  afgeleide laten berekenen. Het commando

`diff( exp(-x), x$n );`

geeft als uitvoer

$$(-1)^n e^{-x}$$

Probeer eens van tevoren te bedenken wat het resultaat is van `diff( sin(t), t$n )`.

<sup>12</sup>Mits aan de voorwaarden van de impliciete functiestelling is voldaan.

**Inerte en actieve vorm.** Het commando `diff( )` berekent direct de afgeleide. We noemen dat de *actieve vorm* van de afgeleide. Dit commando heeft ook een zogenaamde *inerte vorm*, `Diff`, dus geschreven met een hoofdletter. De *inerte vorm* `Diff(x^2, x)` berekent de afgeleide niet, en geeft dus niet  $2x$  als antwoord, maar  $\frac{d}{dx}(x^2)$ . Het commando `value` voert dan alsnog de berekening uit. Meestal wordt de *inerte vorm* gebruikt om een prettiger leesbaar werkblad te krijgen of om gemakkelijker te kunnen controleren of de te differentiëren functie goed is ingetypt.

**D** **Functionies.** Voor de *procedure* `f` is het commando `D` bedoeld. Als `f` de *procedure* `t -> t^2` is, dan is de afgeleide `D(f)` ook een *procedure*, namelijk `t -> 2*t`. De *functiewaarde* in het punt `x` krijgen we in dat geval dus met `D(f)(x)`. Omdat `D` zelf een *procedure* is, kunnen we voor de tweede afgeleide `(D@@D)(f)` en voor de derde afgeleide `(D@@@)(f)` gebruiken enzovoort (zie §3.2).

Als `f` een functie is, dan is `f(x)` een *expressie*. Daarom is ook `diff(f(x), x)` mogelijk, maar dat levert uiteraard geen *procedure*, maar een *expressie*.

## Voorbeeldopgave

Gegeven  $f : x \mapsto \sin(x) \cos(2x)$ . Bereken de afgeleide  $f'(3)$ .

## Voorbeeldsessie

### (1) Met $f$ als *expressie*

We definiëren de functie eerst als de *expressie* `fx`:

```
> fx := sin(x) * cos(2*x);
fx := sin(x) cos(2 x)
```

De *inerte vorm* (met hoofdletter):

```
> Diff(fx,x); value(%);
d
-- (sin(x) cos(2 x))
cos(x) cos(2 x) - 2 sin(x) sin(2 x)
```

De *actieve vorm* (met kleine letter) berekent de afgeleide direct:

```
> df := diff(fx,x);
df := cos(x) cos(2 x) - 2 sin(x) sin(2 x)
```

De afgeleide  $f'(3)$  moet in dit geval worden uitgerekend door  $x = 3$  in deze uitdrukking te substitueren:

```
> df_3 := subs( x=3, df );
df_3 := cos(3) cos(6) - 2 sin(3) sin(6)
```

of:

```
> fd_3 := eval( df, x=3 );
fd_3 := cos(3) cos(6) - 2 sin(3) sin(6)
Van de expressie df kunnen we een functie maken met
> Df := unapply( df, x );
Df := x → cos(x) cos(2 x) - 2 sin(x) sin(2 x)
De afgeleide in x = 3 wordt dan eenvoudig:
> Df(3);
cos(3) cos(6) - 2 sin(3) sin(6)

(2) Met f als procedure
We definiëren f als de procedure f:
> f := t -> sin(t) * cos(2*t);
f := t → sin(t) cos(2 t)
> D(f);
t → cos(t) cos(2 t) - 2 sin(t) sin(2 t)
> D(f)(3);
cos(3) cos(6) - 2 sin(3) sin(6)
```

## Toelichting

Het resultaat verkregen met `diff` is altijd een Maple-*expressie*! We kunnen  $f'(3)$  in dat geval dus alleen uitrekenen door `subs` of `eval` te gebruiken.

De afgeleide van een *functie* (dat wil zeggen een Maple-procedure) wordt verkregen met de functie `D`; het resultaat is weer een Maple-procedure. Let op dat er in `D(f)(3)` geen overbodige haakjes staan: de functie `D` wordt toegepast op de functie `f` en moet dus als `D(f)` worden genoteerd. Het resultaat is de functie  $f'$ . Voor de functiewaarde  $f'(3)$  moet de functie `D(f)` worden toegepast op het argument 3, we noteren `D(f)(3)`.  $\diamond$

In de volgende voorbeeldsessie gaan we in op een aantal veel gemaakte fouten.

## Voorbeeldsessie

```
> fx := sin(x) * cos(2*x);
> df := diff(fx,x);
df := cos(x) cos(2 x) - 2 sin(x) sin(2 x)
> df(3);
cos(x)(3) cos(2 x)(3) - 2 sin(x)(3) sin(2 x)(3)
Dit is niet  $f'(3)$  — let op de toegevoegde (3)'s
> Df := x -> df;
Df := x → df
```

```
> Df(3);
                                cos(x) cos(2 x) - 2 sin(x) sin(2 x)
> Df := x -> diff(fx,x);
                                Df := x ->  $\frac{\partial}{\partial x} f x$ 
> Df(3);

Error, (in Df) invalid input: diff received 3, which is not
valid for its 2nd argument
Niet fout, maar een omweg:
> f := x -> sin(x) * cos(2*x):
> Df := unapply( diff(f(x),x), x );
                                Df := x -> cos(x) cos(2 x) - 2 sin(x) sin(2 x)
```

## Toelichting

Vaak zal men de neiging hebben om een *expressie* **df** te *gebruiken* alsof het een *functie* is. Maple geeft dan in het algemeen géén foutmelding<sup>13</sup>. Aan de uitvoer is echter gemakkelijk te zien dat er iets niet in de haak is.

unapply

De pijltjesmethode van Module 3 is blijkbaar niet zonder meer geschikt om van de berekende *expressie* **df** een *functie* te maken; we hebben daarvoor het commando **unapply** nodig. In het eerste geval (**Df := x -> df**) weigert Maple om de waarde  $x = 3$  in te vullen. In het tweede geval (**Df := x -> diff(f(x),x)**) interpreteert Maple de aanroep **Df(3)** als **diff(f(3),3)** en geeft een foutmelding. In Module 7 zullen we nader ingaan op het verschil tussen **->** en **unapply**. Hanteer vooralsnog de volgende vuistregel:



Gebruik **->** om een functie te definiëren als u rechts van de pijl een volledig uitgeschreven formule kunt plaatsen;  
Gebruik **unapply** als u een functie wilt maken van een door Maple berekende expressie.

Het laatste statement van deze voorbeeldsessie is eigenlijk een curiositeit. We gaan eens precies, van binnen naar buiten, na wat hier wordt berekend.

- Van de *functie* **f** wordt de *expressie* **f(x)** gemaakt;
- deze wordt gedifferentieerd; het resultaat is de *expressie* **diff(f(x),x)**;
- hier wordt dan ten slotte met **unapply** weer een *functie* van gemaakt.

<sup>13</sup>Omdat uitdrukkingen als ‘cos(x)’ en ‘sin(2x)’ ook als namen van functies mogen worden gebruikt.

Uiteindelijk is dit dus een ingewikkelde manier om  $Df := D(f)$  te schrijven.  $\diamond$

**Partiële afgeleiden.** Als  $f$  een functie van meer variabelen is, kan  $D$  ook worden gebruikt voor het berekenen van *partiële afgeleiden*. Stel  $f : \mathbb{R}^2 \rightarrow \mathbb{R}$  is gedefinieerd met

$$f := (x, y) \rightarrow x^2 y - 3 y^2;$$

dan levert  $D[1](f)$  de partiële afgeleide naar de *eerste* variabele ( $x$ ), dus in dit voorbeeld de procedure  $(x, y) \rightarrow 2xy$ , en  $D[2](f)$  de partiële afgeleide naar  $y$ . Hogere orde partiële afgeleiden verkrijgt men met bijvoorbeeld  $D[2,1](f)$  voor  $\frac{\partial^2 f}{\partial y \partial x}$  enzovoort.

Uiteraard geeft `diff(expressie, x)` ook een partiële afgeleide, maar in de vorm van een expressie in plaats van een procedure. Zie verder Module 16.

**Impliciet differentiëren.** Hiervoor heeft Maple de opdracht `implicitdiff`. We lichten het gebruik toe aan de hand van een voorbeeld.

### Voorbeeldopgave

Bereken  $y'(x)$  als  $y(x)$  wordt gedefinieerd door de vergelijking

$$2xy^3 + 3x^2y^2 = 1$$

### Voorbeeldsessie

```
> f := (x,y) -> 3*x*y^3 + 3*x^2*y^2;
```

$$f := (x, y) \rightarrow 3xy^3 + 3x^2y^2$$

```
> implicitdiff( f(x,y)=1, y, x );
```

$$-\frac{y(y+2x)}{x(3y+2x)}$$

Controle:

```
> df := diff( f(x,y(x))=1, x );
```

$$\begin{aligned} df &:= 3y(x)^3 + 9xy(x)^2 \left(\frac{d}{dx}y(x)\right) + 6xy(x)^2 \\ &\quad + 6x^2y(x) \left(\frac{d}{dx}y(x)\right) = 0 \end{aligned}$$

```
> 'y'(x)' := solve( df, diff(y(x),x) );
```

$$y'(x) := -\frac{y(x)(y(x)+2x)}{x(3y(x)+2x)}$$

## Toelichting

De procedure `implicitdiff` heeft drie argumenten nodig: de vergelijking, de afhankelijke variabele (dat is de impliciet gedefinieerde functie) en de onafhankelijke variabele.

Merk op dat Maple een formule met een onbekende functie erin ( $y(x)$  in dit geval) gewoon kan differentiëren. Zie verder §6.9. We kunnen ook een variabele met een ingewikkelde naam (`diff(y(x), x)` in dit geval) uit een vergelijking oplossen.  $\diamond$

Met `implicitdiff` kunnen ook meerdere (partiële) afgeleiden van door een stelsel vergelijkingen impliciet gedefinieerde functies worden berekend. Raadpleeg `?implicitdiff` voor meer informatie.

## 6.4 Enkele integratieregels

Als  $F'(x) = f(x)$  voor alle  $x$ , dan heet  $F$  een *primitieve* van  $f$ , notatie:  $F = \int f$ , of:  $F(x) = \int f(x) dx$ . We herhalen eerst de volgende stellingen:

**Hoofdstelling:** Als  $F$  een primitieve is van  $f$ , dan is

$$\int_a^b f(x) dx = \left[ F(x) \right]_a^b = F(b) - F(a).$$

**Partiële Integratie:** Zij  $I$  een interval. Als  $F : I \rightarrow \mathbb{R}$  en  $G : I \rightarrow \mathbb{R}$  primitieven zijn van de functies  $f$  en  $g$ , dan is

$$\int_a^b f(t)G(t) dt = \left[ F(t)G(t) \right]_a^b - \int_a^b F(t)g(t) dt.$$

**Substitutieregel:** Zij  $I$  een interval, en  $f : I \rightarrow \mathbb{R}$  continu differentieerbaar, en stel dat de inverse  $f^{-1}$  bestaat. Als  $g : J \rightarrow \mathbb{R}$  continu is, en  $J \subset f(I)$ , dan is

$$\int_a^b g(t) dt = \int_{f^{-1}(a)}^{f^{-1}(b)} g(f(u)) f'(u) du.$$

We zeggen dat we  $f(u)$  voor  $t$  hebben gesubstitueerd.

## 6.5 Integreren

`int`  
`Int`

Met de opdracht `int(fx,x)` wordt een primitieve van  $f$  bepaald. Hierin is  $fx$  een expressie waarin  $x$  voorkomt. De inerte vorm is `Int`. De bepaalde integraal

$$\int_a^b f(x) dx$$

wordt berekend via

```
int(fx, x=a..b);
```

en met `int(f(t), t=a..b)` of `int(f, a..b)` als  $f$  als procedure is gedefinieerd.<sup>14</sup>

Het kan voorkomen dat Maple geen gesloten uitdrukking voor de primitieven van  $f$  kan berekenen. In dat geval geeft het commando `int` de integraal zonder resultaat in een gestileerde vorm terug, dus in dezelfde vorm als wanneer u `Int` had gegeven. Dat zal maar zelden gebeuren. Voor veel van dit soort integralen heeft Maple wel de een of andere exotische functie, die in feite niet veel anders is dan de naam voor een primitieve.

Vooral bij het berekenen van integralen moet men overigens steeds opletten of de door Maple geproduceerde resultaten correct zijn. Zie opgave 6.4.

**Numerieke integratie.** Als er geen gesloten of geen bruikbare uitdrukking is voor de (bepaalde!) integraal, dan kan geprobeerd worden hiervan een *numerieke benadering* te vinden met het commando `evalf` (Zie Module 3):

```
evalf(Int(f(x), x=a..b));
```

Merk op dat hier de *inerte vorm* van het integratiecommando gebruikt is. Als de actieve vorm gebruikt zou worden, dan probeert Maple eerst de exacte waarde te bepalen en deze vervolgens numeriek te benaderen. Bij de inerte vorm gaat Maple direct een numerieke integratiemethode toepassen. Dit is meestal sneller en vaak ook nauwkeuriger. (Zie de voorbeeldsessie.)

<sup>14</sup>Voor de *onbepaalde* integraal is er niet zo'n commando zonder parameter. Men moet dan `int(f(t),t)` gebruiken.

## Voorbeeldopgave

Bereken  $\int_0^1 e^{-x^2} x^{-\frac{1}{2}} dx$ .

## Voorbeeldsessie

```
> st := time(): antw := int( exp(-x^2)*x^(-1/2), x=0..1 );

          8
          e(-1/2) WhittakerM( $\frac{1}{8}, \frac{5}{8}, 1$ )
          + 2 e(-1/2) WhittakerM( $\frac{9}{8}, \frac{5}{8}, 1$ )
> evalf(antw); ‘benodigde tijd’ = time()-st;

1.689677190
benodigde tijd = 0.256

> restart;
> st := time(): Int( exp(-x^2)*x^(-1/2), x=0..1 ): % = evalf(%);
‘benodigde tijd’ = time()-st;

 $\int_0^1 \frac{e^{(-x^2)}}{\sqrt{x}} dx = 1.689677190$ 
benodigde tijd = 0.052
```

## Toelichting

Ook oneigenlijke integralen gaan kennelijk goed! (De integrand is singulier in  $x = 0$ ). De numerieke benadering die men verkrijgt door het ‘exacte’ antwoord (met de Whittakerfuncties erin) te benaderen kost aanzienlijk meer tijd dan de directe numerieke integratie. In dit geval is die kwartseconde natuurlijk nog wel overkomelijk, maar als men bijvoorbeeld een grafiek wil laten tekenen van een functie met een integraaluitdrukking erin, dan moet er honderden keren zo’n integraal worden berekend.

Het antwoord is positief, en dat is ook te verwachten omdat de integrand positief is.  $\diamond$

## 6.6



## Partiële integratie en substitutie

Integration-  
Tools

De bibliotheek `IntegrationTools` bevat een aantal hulpmiddelen om met ‘inerte’ integralen (dus van het type `Int`) te manipuleren.

**Partiële integratie.** We kunnen de integraal  $\int_a^b f(t)G(t) dt$  partiëel integreren (zoals in de stelling) door het commando

**Parts**

```
Parts( Int(f(t)*G(t), t=a..b), G(t) );
```

dat in de **IntegrationTools**-bibliotheek aanwezig is. Als tweede argument van **Parts** wordt dus de functie  $G$  meegegeven die moet worden gedifferentieerd.

### Voorbeeldopgave

Bewijs, door tweemaal partiëel integreren, dat

$$\int_0^{\pi/2} x^n \sin(x) dx = n \left(\frac{\pi}{2}\right)^{(n-1)} - n(n-1) \int_0^{\pi/2} x^{(n-2)} \sin(x) dx.$$

### Voorbeeldsessie

```
> p := Int( x^n*sin(x), x=0..Pi/2 );
```

$$p := \int_0^{1/2 \pi} x^n \sin(x) dx$$

We moeten de graad van  $x^n$  verlagen, dus  $x^n$  moet worden gedifferentieerd:

```
> with(IntegrationTools):
```

```
> p1 := Parts( p, x^n );
```

$$p1 := - \lim_{x \rightarrow 0^+} \cos(x) x^n - \int_0^{1/2 \pi} \frac{\cos(x) x^n n}{x} dx$$

```
> p2 := simplify(p1) assuming n::posint;
```

$$p2 := n \int_0^{1/2 \pi} \cos(x) x^{-1+n} dx$$

```
> Parts( p2, x^(n-1) );
```

$$n \left( 2 \frac{\pi^n}{2^n \pi} - \lim_{x \rightarrow 0^+} \sin(x) x^{-1+n} - \int_0^{1/2 \pi} \frac{\sin(x) x^{-1+n} (-1+n)}{x} dx \right)$$

```
> p3 := Parts( p2, x^(n-1) ) assuming n>1;
```

$$p3 := n \left( 2 \frac{\pi^n}{2^n \pi} - \int_0^{1/2 \pi} \frac{\sin(x) x^{-1+n} (-1+n)}{x} dx \right)$$

```
> expand(p3);
```

$$2 \frac{n \pi^n}{2^n \pi} - n(-1+n) \int_0^{1/2 \pi} \frac{x^n \sin(x)}{x^2} dx$$

```
> map(simplify,expand(p3)) assuming n>1;
```

$$2^{1-n} n \pi^{-1+n} - n(-1+n) \int_0^{1/2 \pi} \sin(x) x^{-2+n} dx$$

## Toelichting

We zien dat Maple zeer zorgvuldig is: bij de berekening van `p1` (eerste keer partieel integreren) vult het nog geen  $x = 0$  in in  $x^n$ , maar wacht daarmee totdat wij aangeven dat  $n$  een positief geheel getal is. Bij de tweede keer partieel integreren vermelden we dat  $n > 1$  is.

Met behulp van het `map`-commando<sup>15</sup> worden beide termen van `p3` afzonderlijk vereenvoudigd. Het heeft geen zin om te proberen de laatste uitdrukking met Maple precies in de vorm van het rechterlid in de opgave te krijgen. We kunnen hieraan direct wel zien dat het klopt.  $\diamond$

**Opmerking.** Als we bijvoorbeeld alléén de procedure `Parts` uit de bibliotheek `IntegrationTools` hadden willen gebruiken, dan had dat ook gekund met `use IntegrationTools in Parts(p,x^n) end use`; Een andere mogelijkheid is gebruik te maken van de operator `:-` (waarmee een procedure uit een zogenaamde ‘module’ wordt aangegeven):

```
IntegrationTools:-Parts(p,x^n);
```

Deze operator werkt trouwens niet bij elke bibliotheek, maar alleen bij de nieuwere (of vernieuwde) bibliotheken. Als hij niet werkt, dan is de bibliotheek waarschijnlijk verouderd, en vervangen door een nieuwere versie.

Change

**Substitutie.** Gegeven de integraal  $\int_a^b g(t)dt$ . De substitutie  $t = f(u)$  wordt in Maple uitgevoerd via `Change`, eveneens uit de bibliotheek `IntegrationTools`:

```
Change( Int(g(t), t=a..b), t=f(u));
```

De grenzen veranderen automatisch mee!

## Voorbeeldopgave

Bewijs:

$$\int_0^1 \frac{1}{1-x^2} dx = \int_0^{\pi/2} \frac{1}{\cos(u)} du.$$

## Voorbeeldsessie

```
> with(IntegrationTools):
> p := Int( 1/(1-t^2), t=0..1 );
```

<sup>15</sup>Zie Module 8 voor meer informatie over `map`

$$p := \int_0^1 \frac{1}{1-t^2} dt$$

```
> Change( p, t=sin(u) );
```

$$\int_0^{\frac{\pi}{2}} \frac{1}{\cos(u)} du$$

## 6.7 Limietberekening

limit  
Limit

De *limiet* van  $f$  voor  $x$  naderend naar  $b$ ,  $\lim_{x \rightarrow b} f(x)$ , wordt berekend met de opdracht `limit(f(x), x=b)`. Ook deze functie heeft een inerte vorm: `Limit`.

### Voorbeeldopgave

Bereken  $\lim_{x \rightarrow 0} \frac{\sin x}{x}$ .

### Voorbeeldsessie

```
> f := x -> sin(x)/x;
```

$$f := x \rightarrow \frac{\sin(x)}{x}$$

Inerte vorm:

```
> Limit( f(x), x=0 ): % = value(%);
```

$$\lim_{x \rightarrow 0} \frac{\sin(x)}{x} = 1$$

Aktieve vorm:

```
> limit( f(x), x=0 );
```

1

### Toelichting

Ondanks het feit dat zowel de teller als de noemer 0 opleveren in de limiet, beweert Maple dat de limiet van het quotiënt bestaat en gelijk is aan 1 (hetgeen uiteraard correct is).

In deze voorbeeldsessie kunt u ook zien hoe de inerte vorm `Limit`, gecombineerd met `value` een fraaie uitvoer oplevert (denk aan de dubbele punt na het eerste statement).  $\diamond$

## 6.8



## Sommatie

sum, Sum

add

Voor *sommatie* zijn in Maple twee commando's beschikbaar: `sum` (inerte vorm: `Sum`) en `add` (geen inerte vorm). Het gebruik van deze commando's voor de sommatie is precies analoog aan dat van `int`.

Het commando `sum` kan (ook) gebruikt worden voor *symbolische* berekeningen, men kan hiermee ook een som voor  $k = 1$  tot  $n$  berekenen: als de bovengrens  $n$  geen waarde heeft, komt er een formule uit met  $n$  erin (vergelijk de primitieve). Ook oneindige sommen kunnen met `sum` worden berekend.

Met het commando `add` plaatst Maple alléén plustekens tussen de termen. Dat betekent dat de onder- en bovengrens van de sommatie-index alleen vaste natuurlijke getallen mogen zijn. Vaak kan ook `sum` in plaats van `add` worden gebruikt; echter, als er andere dingen dan getallen bij elkaar moeten worden opgeteld (bijvoorbeeld matrices), dan moet `add` worden gebruikt.

We laten het zien aan de hand van een voorbeeld.

### Voorbeeldsessie

```
> Sn := Sum( 1/k^2, k=1..10): % = value(%);
```

$$\sum_{k=1}^{10} \frac{1}{k^2} = \frac{1968329}{1270080}$$

```
> Sum( 1/k^2, k=1..infinity): % = value(%);
```

$$\sum_{k=1}^{\infty} \frac{1}{k^2} = \frac{\pi^2}{6}$$

```
> k := 12: sum( 1/k^2, k=1..10);
```

Error, (in sum) summation variable previously assigned, second argument evaluates to 12 = 1 .. 10

```
> k := 12: add( 1/k^2, k=1..10);
```

$$\frac{1968329}{1270080}$$

```
> k := 12: add( 1/k^2, k=1..infinity);
```

Error, unable to execute add

```
> k := 12: sum( '1/k^2', 'k'=1..infinity);
```

$$\frac{\pi^2}{6}$$

## Toelichting

Pas op dat in `sum` de sommatie-index geen waarde mag hebben; zie de foutmelding. Sommigen maken er een gewoonte van om in `sum` *altijd* quotes te gebruiken, waardoor een eventuele waarde van de sommatie-index wordt verwijderd. In `add` is de sommatie-index automatisch een dummy. Maar `add` kan weer niet ‘symbolisch’ rekenen: voor de som tot  $\infty$  is `sum` nodig.  $\diamond$

## 6.9 Ongedefinieerde functies

Een uitdrukking als  $f(x)$  kan in het eerste argument van `diff`, `int` of `lim` worden gebruikt, óók als  $f$  niet van tevoren is gedefinieerd als `f := x -> ...`. Zo’n ‘ongedefinieerde functie-aanroep’  $f(x)$  is van het Maple-type *function*, terwijl  $f$  zelf van het type *procedure* is. Zie verder Module 26.

### Voorbeeldsessie

```
> df := diff( f(x), x );
```

$$df := \frac{d}{dx} f(x)$$

```
> df1 := subs( x=1, df ); df1;
```

$$df1 := \text{diff}(f(1), 1)$$

Error, invalid input: diff received 1, which is not valid for its 2nd argument

```
> df1 := eval( df, x=1 );
```

$$df1 := \left( \frac{d}{dx} f(x) \right) \Big|_{x=1}$$

```
> f := x -> a*x^2 + b*x + c;
```

$$f := x \rightarrow ax^2 + bx + c$$

```
> df1;
```

$$2a + b$$

## Toelichting

Zolang Maple niet weet wat  $f$  is, doet het alsof de inerte vorm `Diff` is gebruikt. In deze sessie zien we ook een interessant verschil tussen `subs` en `eval`: `subs(x=1,df)` vervangt botweg elke  $x$  door een 1 in `diff(f(x),x)`, waardoor een onzinnige uitdrukking ontstaat. Het

commando `eval` is in zoverre ‘intelligent’, dat het constateert dat  $x=1$  in `diff(f(x),x)` pas kan worden ingevuld als  $f'(x)$  bekend is.  $\diamond$

Ook op niet gedefinieerde functies kan Maple de bekende rekenregels toepassen.

### Voorbeeldsessie

```
> F := f@g;
> diff(F(x),x);
```

$$D(f)(g(x)) \left( \frac{d}{dx} g(x) \right)$$

```
> D(f@g)(x);
```

$$D(f)(g(x)) D(g)(x)$$

```
> diff( f(x)*g(x), x );
```

$$\left( \frac{d}{dx} f(x) \right) g(x) + f(x) \left( \frac{d}{dx} g(x) \right)$$

```
> D(f*g);
```

$$D(f) g + f D(g)$$

```
> D(f*g)(x);
```

$$D(f)(x) g(x) + f(x) D(g)(x)$$

```
> diff( f(x)*g(x)*h(x), x$4 );
```

(output niet afgedrukt)

```
> Limit( x/f(x), x=0 ): % = value(%);
```

$$\lim_{x \rightarrow 0} \frac{x}{f(x)} = 0$$

### Toelichting

In deze sessie is te zien dat Maple de kettingregel en de productregel kent. Zodra Maple een uitdrukking als `f(x)`, `f@g` of `D(f)` tegenkomt handelt het alsof `f` een procedure is.

Uit het voorbeeld met de limiet blijkt trouwens dat men moet oppassen met dergelijke ongedefinieerde zaken: het antwoord hoeft niet juist te zijn als  $f(0) = 0$ .  $\diamond$

### Opgave 6.1

Bereken de  $6^{\text{e}}$  afgeleide van  $x^2 e^x$ .

Geef een algemene formule voor de  $n^{\text{de}}$  afgeleide en laat zien dat deze klopt voor  $n = 6$ .

### Opgave 6.2

Bereken een primitieve van  $\frac{x^2 + 2x + 3}{x^2 + 4x + 6}$ .

### Opgave 6.3

Bereken  $\int_{-\infty}^{\infty} e^{-x^2} dx$ .

### Opgave 6.4

Gegeven is de functie  $f_a : x \mapsto \frac{1}{(x-a)^2}$  en de integraal

$$A_a = \int_{-1}^1 f_a(x) dx.$$

- Definieer  $f_a$  als  $\mathbf{f} := (\mathbf{a}, \mathbf{x}) \rightarrow \dots$  (dus als functie van *twee* variabelen);
- Bepaal een primitieve  $F_a(x)$  van  $f_a(x)$ , en met behulp hiervan de integraal  $A_a$ . Klopt het antwoord voor elke waarde van  $a$ ?
- Bereken  $A_a$  als functie van  $a$  als bepaalde integraal van  $\mathbf{f}(\mathbf{a}, \mathbf{x})$  (dus het antwoord moet de vorm  $\mathbf{A} := \mathbf{a} \rightarrow \dots$  hebben) en met behulp daarvan  $A_{10}$ ,  $A_0$  en  $A_1$ ;
- Bereken ten slotte  $A_{10}$ ,  $A_0$  en  $A_1$  rechtstreeks, door  $\mathbf{f}(10, \mathbf{x})$  enzovoort te integreren.

### Opgave 6.5

Bewijs, door tweemaal partieel integreren, dat

$$\int_0^{\pi/2} x^n \cos(x) dx = \left(\frac{\pi}{2}\right)^n - n(n-1) \int_0^{\pi/2} x^{(n-2)} \cos(x) dx.$$

### Opgave 6.6

Zie ook Module 4. Bewijs met de substitutie  $u = \tan(t/2)$  dat

$$\int_0^{\pi/2} \frac{1}{4 + 5 \cos(t)} dt = -2 \int_0^1 \frac{1}{(u-3)(u+3)} du.$$

### Opgave 6.7

Bereken  $\lim_{x \rightarrow 0} \frac{\cos x - 1}{x^2}$ .

### Opgave 6.8

Bereken  $\lim_{n \rightarrow \infty} \left(1 + \frac{x}{n}\right)^n$ .

### Opgave 6.9

Bereken  $\lim_{x \rightarrow \infty} x e^{-x}$  en  $\lim_{x \rightarrow \infty} x^n e^{-x}$ . Bereken ook  $\lim_{x \rightarrow -\infty} x^n e^{-x}$ . Kloppen de antwoorden voor elke  $n$ ?

### Opgave 6.10

Raadpleeg ?limit en bepaal

$$\lim_{x \uparrow 0} \frac{\sin \sqrt{-x}}{\tan \sqrt{-x/2}}.$$

### Opgave 6.11

Gegeven  $s_n = \sum_{k=1}^n \frac{1}{k} - \ln n$ .

- (a) Bereken  $s_n$ ;
- (b) Bepaal  $\lim_{n \rightarrow \infty} s_n$ ;
- (c) Geef ook een numerieke benadering van  $s$ .

